Bericht und weitere Dateien als Anhang versenden
Versionen: Access 2007, 2003, 2002/XP und 2000

Mit Access soll ein Bericht per E-Mail verschickt werden, was beispielsweise mit dem folgenden Befehl gelingt:

DoCmd.SendObject acReport, stDocName, acFormatRTF, , , ,
stSubject, stMessageText

Problematisch wird es, wenn an die E-Mail zusätzlich noch eine Datei angehängt werden soll. Mit "DoCmd.SendObject" lässt sich das leider nicht lösen. Sie können aber Outlook per Office Automation fernsteuern, eine Mail anlegen und dann beliebige Anhänge zuordnen. Wechseln Sie zunächst in die VBA-Entwicklungsumgebung, wählen Sie das Menü Extras-Verweise an und setzen Sie eine Referenz auf "Microsoft Outlook x.0 Object Library" (x.0= 12.0 für Outlook 2007, 11.0 für Outlook 2003, 10.0 für Outlook 2002 oder 9.0 für Outlook 2000). Außerdem benötigen Sie eine Schaltfläche mit folgender Ereignisprozedur "Beim Klicken":

Private Sub btnSendReports_Click()
  Dim objOutl As Outlook.Application
  Dim objMail As Outlook.MailItem

  DoCmd.OutputTo acOutputReport, _
                 "Preisliste", acFormatRTF, _
                 "Z:\Test\Preisliste.rtf"
  DoCmd.OutputTo acOutputReport, _
                 "Katalog", acFormatRTF, _
                 "Z:\Test\Katalog.rtf"
  Set objOutl = CreateObject("Outlook.Application")
  Set objMail = objOutl.CreateItem(olMailItem)
  With objMail
    .Subject = "Betreff..."
    .Body = "Nachrichtentext..."
    .Attachments.Add "Z:\Test\Preisliste.rtf"
    .Attachments.Add "Z:\Test\Katalog.rtf"
    .Attachments.Add "Z:\Test\Datei.doc"
    .Attachments.Add "Z:\Test\Datei.ppt"
    .Display
  End With

End Sub

Diese Anweisungen exportieren zunächst zwei Berichte - Preisliste und Katalog - im RTF-Format. Dann wird eine Verbindung zu Outlook aufgebaut und eine neue Nachricht erstellt. Dieser Nachricht ordnen wir dann Betreff und Nachrichtentext sowie vier Dateien - Preisliste, Katalog, Word- und PowerPoint-Datei - als Anhang zu. Abschließend wird die Nachricht angezeigt und kann dann nach Festlegung des Empfängers auf die Reise geschickt werden.
